

01

```
PREFIX i: <http://nosql.opendata.cz/school/terms#>
SELECT ?n ?f ?l
FROM <http://nosql.opendata.cz/school/pdb-251/>
WHERE {
    ?s rdf:type i:Student .
    ?s i:number ?n .
    ?s i:name ?b .
    ?b i:first ?f .
    ?b i:last ?l .
}

... WHERE {
    ?s rdf:type i:Student ;
        i:number ?n ;
        i:name ?b .
    ?b i:first ?f ; i:last ?l .
}

... WHERE {
    ?s rdf:type i:Student ;
        i:number ?n ;
        i:name [ i:first ?f ; i:last ?l ] .
}

... WHERE {
    ?s rdf:type i:Student ;
        i:number ?n ;
        i:name _:b .
    _:b i:first ?f ; i:last ?l .
}

... SELECT $n $f $l
... WHERE {
    $s rdf:type i:Student ;
        i:number $n ;
        i:name [ i:first $f ; i:last $l ] .
}
```

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
SELECT ?c ?t
FROM <http://nosql.opendata.cz/school/pdb-251/>
WHERE {
    ?p rdf:type i:Course ;
        i:code ?c ;
        i:title ?t .
    FILTER ( regex(?c, "^NPRG") )
}
ORDER BY ?t

... WHERE {
    ?p rdf:type i:Course ;
        i:code ?c .
    FILTER ( regex(?c, "^NPRG") )
    ?p i:title ?t .
}

... FILTER regex(?c, "^NPRG") ...

... ORDER BY ASC(?t)

... [ rdf:type i:Course ; i:code ?c ; i:title ?t . ] ...

... [] rdf:type i:Course ; i:code ?c ; i:title ?t . ...

... _: rdf:type i:Course ; i:code ?c ; i:title ?t . ...

```

03

```
PREFIX i: <http://nosql.opendata.cz/school/terms#>
SELECT ?n ?e
FROM <http://nosql.opendata.cz/school/pdb-251/>
WHERE {
    ?s rdf:type i:Student ;
        i:number ?n .
    OPTIONAL { ?s i:email ?e . }
}
```

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
SELECT ?n ?e ?w
FROM <http://nosql.opendata.cz/school/pdb-251/>
WHERE {
    ?s rdf:type i:Student ;
        i:number ?n .
    OPTIONAL { ?s i:email ?e }
    OPTIONAL { ?s i:web ?w }
}

```

Incorrect:

```

... WHERE {
    ?s rdf:type i:Student ;
        i:number ?n
    OPTIONAL { ?s i:email ?e ; i:web ?w }
}

```

Incorrect:

```

... WHERE {
    ?s rdf:type i:Student ;
        i:number ?n .
    OPTIONAL {
        ?s i:email ?e
        OPTIONAL { ?s i:web ?w }
    }
}

```

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT DISTINCT ?p ?c
FROM <http://nosql.opendata.cz/school/pdb-251/>
WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  ?e i:course ?p ; i:day ?d ; i:semester s:semester251 .
  FILTER ( (?d = "MON") || (?d = "FRI") )
}

... WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  [ i:course ?p ; i:day ?d ; i:semester s:semester251 ]
  FILTER ( (?d = "MON") || (?d = "FRI") )
}

... WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  ?e i:course ?p ; i:day ?d ; i: semester ?s .
  FILTER ( (?d = "MON") || (?d = "FRI") )
  FILTER ( ?s = s:semester251 )
}

... WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  ?e i:course ?p ; i:day ?d ; i: semester ?s .
  ?s i:year "2025/26" ; i:period "winter" .
  FILTER ( (?d = "MON") || (?d = "FRI") )
}

... WHERE {
  {
    ?p rdf:type i:Course ; i:code ?c .
    ?e i:course ?p ; i:day ?d ; i:semester s:semester251 .
    FILTER ( ?d = "MON" )
  } UNION {
    ?p rdf:type i:Course ; i:code ?c .
    ?e i:course ?p ; i:day ?d ; i:semester s:semester251 .
    FILTER ( ?d = "FRI" )
  }
}

... WHERE {
  {
    ?p rdf:type i:Course ; i:code ?c .
    ?e i:course ?p ; i:day "MON" ; i:semester s:semester251 .
  } UNION {
    ?p rdf:type i:Course ; i:code ?c .
    ?e i:course ?p ; i:day "FRI" ; i:semester s:semester251 .
  }
}

```

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT ?p ?c
FROM <http://nosql.opendata.cz/school/pdb-251/>
WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  FILTER NOT EXISTS {
    ?e i:course ?p ; i:day ?d ; i:semester s:semester251 .
    FILTER ( (?d = "MON") || (?d = "FRI") )
  }
}

```

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT ?p ?c
FROM <http://nosql.opendata.cz/school/pdb-251/>
WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  MINUS {
    ?e i:course ?p ; i:day ?d ; i:semester s:semester251 .
    FILTER ( (?d = "MON") || (?d = "FRI") )
  }
}

```

Incorrect:

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT DISTINCT ?p ?c
FROM <http://nosql.opendata.cz/school/pdb-251/>
WHERE {
  ?p rdf:type i:Course ; i:code ?c .
  ?e i:course ?p ; i:day ?d ; i:semester s:semester251 .
  FILTER ( (?d != "MON") && (?d != "FRI") )
}

```

```

PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT ?p ?c
FROM <http://nosql.opendata.cz/school/pdb-251/>
WHERE {
    ?p rdf:type i:Course ; i:code ?c .
    OPTIONAL {
        ?e i:course ?p ; i:day ?d ; i:semester s:semester251 .
        FILTER ( (?d = "MON") || (?d = "FRI") )
    }
    FILTER ( ! bound(?e) )
}

... WHERE {
    ?p rdf:type i:Course ; i:code ?c .
    OPTIONAL {
        ?e1 i:course ?p ; i:day "MON" ; i:semester s:semester251
    }
    OPTIONAL {
        ?e2 i:course ?p ; i:day "FRI" ; i:semester s:semester251
    }
    FILTER ( ! bound(?e1) && ! bound(?e2) )
}

```

```
PREFIX i: <http://nosql.opendata.cz/school/terms#>
PREFIX s: <http://nosql.opendata.cz/school/>
SELECT ?f ?l (AVG(?r) AS ?a)
FROM <http://nosql.opendata.cz/school/pdb-251/>
WHERE {
    ?s rdf:type i:Student ;
        i:name [ i:first ?f ; i:last ?l ] ;
        i:enroll [ i:course ?c ; i:result ?r ; i:semester s:semester251 ] .
}
GROUP BY ?s ?f ?l
HAVING (COUNT(?c) <= 10)
```